

An Overview of Two Approaches for Knowledge Representation and Reasoning in the Context of Planning

Gregory Gelfond and Glen Hunt

December 8, 2010

Contents

1	Introduction	2
2	Classical Planning Overview	2
2.1	Background	2
2.1.1	Difficulties of Classical Planning	2
2.1.2	Representing Classical Planning Problems	3
2.2	PDDL	4
2.3	FF Planner	4
2.4	Planning Domain Examples	5
2.4.1	Blocks World	5
2.4.2	Electrical Circuit	7
2.4.3	Lin's Briefcase	9
2.4.4	Towers of Hanoi	10
3	The Knowledge Representation and Reasoning Approach	11
3.1	The Action Language \mathcal{AL}	11
3.2	The Language of A-Prolog	12
3.3	Domains	13
3.3.1	Lin's Briefcase	13
3.3.2	The Blocks World	16
3.3.3	Electrical Circuit	19
3.3.4	Towers of Hanoi	22
4	Conclusions	25
5	Roles in the Project	27

1 Introduction

Dijkstra viewed programming as the process of the “refinement of specification.” This applies not only to imperative programming, but in the myriad fields of Artificial Intelligence Research. It is especially made clear given that the levels of refinement we are able to obtain in our various formalisms, programs, etc. are inherently tied to the languages we use to reason about our problem domains.

This is also a central view taken by many members of what has been termed: the “Knowledge Representation and Reasoning Community.” Within this community one of the central goals has been to understand and formalize the reasoning of intelligent agents. A number of languages have developed such *action languages*, and the language of *answer set prolog* (denoted throughout as A-Prolog). A rather significant body of work has been done regarding various reasoning tasks such as planning, diagnosis, prediction, counter-factual reasoning, etc., all from the perspective of them being simply one of many reasoning tasks an agent may perform. This has lead to a framework for the design and development of intelligent agents based on a *portfolio* of languages, corresponding roughly to individual levels of refinement about an agent’s knowledge of the domain.

This central view does not appear to be shared by the “Classical Planning” community, given its use of a single language, *PDDL*, which was “originally developed by the AIPS-98 Competition Committee for use in defining problem domains.”

This difference in first principles leads to some interesting consequences which we hope to explore to some extent in this paper. The rest of this paper is structured as follows: an overview of the classical planning approach will be given in the context of four different planning domains, The Blocks World, Towers of Hanoi, Lin’s Briefcase, and an Electrical Circuit. Once this has been, a similar overview of the knowledge representation and reasoning approaches, including descriptions of the action language \mathcal{AL} , followed by the logic programming language of A-Prolog. Once this has been covered, representations of the aforementioned domains will be presented.

2 Classical Planning Overview

2.1 Background

Planning, in general, consists of computing a sequence of actions which, starting in an initial state or states, will achieve a set of goal states. Classical planning, then, is a constrained planning task, which is fully observable, deterministic, finite, static (i.e. the world does not change unless the agent acts), and discrete (i.e. time, actions, objects and effects are not continuous). Nonclassical planning, by contrast, is used to describe planning in partially observable or stochastic environments.

2.1.1 Difficulties of Classical Planning

Typical problem-solving agents which make use of standard search algorithms (i.e. depth-first, breadth-first, A^* , etc) encounter several difficulties when solving classical planning problems; each of these must be kept

in mind when designing a planning agent. First of all, the agent may be overwhelmed by “irrelevant” actions (that is, actions which do not necessarily bring the agent closer to achieving a goal). There are multiple ways within the classical planning community of discouraging a planning agent from considering irrelevant actions; the use of regression search, for example, is one way to attempt to handle this problem.

Another difficulty in classical planning is finding a good heuristic function. Because a problem-solving agent would see the goal state only as a “black box” which is either true or false for a given state, it would require a new heuristic for each specific problem it encounters. A planning agent, however, may have access to a representation of the goal state as a conjunction of subgoals, in which case it can use a single domain-independent heuristic; namely, the number of unsatisfied conjuncts.

Similarly, a problem-solving agent might be inefficient because of its inability to take advantage of problem decomposition. A planning agent which views a state as a conjunction of literals, however, might be able to decompose the goal into a conjunction of subgoals, and then attempt to find a plan to achieve each subgoal. There are tradeoffs to this approach, however. Even if a planning agent can work on subgoals independently, some additional work may be required to combine the resultant subplans. Additionally, for some domains, on one subgoal may undo another.

2.1.2 Representing Classical Planning Problems

Classical planning problems are represented using three formalisms: states, goals and actions. These constructs form the basis of the languages used to encode planning problems in such a way that planners can understand them and operate on them to search for plans.

States in classical planning problems are represented by a conjunction of positive, ground, first-order propositional literals. Classical planning adopts the closed-world assumption, which states that any literals which are not explicitly enumerated as part of the state are false. This state representation may be manipulated by logical inference (when viewed as a conjunction of fluents), or with set operations (when viewed as a set of fluents).

Actions may be represented by an action schema. This schema consists of the action name, a list of all variables used in the schema, a precondition, and an effect. Both the precondition and effect of the action are conjunctions of literals, which may be either positive or negative. The precondition gives a definition of states in which the action can be executed; the effect gives the result of executing the action. More formally, an action a may be executed in some state s if s entails the precondition of a . If an action schema contains variables, there may be multiple states (ground instantiations of the variables in the action) which satisfy the preconditions of the action.

The result of executing some action a in state s is given by state s' . s' is first populated with the set of fluents in s , then all fluents which appear as negative literals in the effects of a are removed, and finally the fluents which appear as positive literals in the effects of a are added.

A set of these action schemas gives the definition of a planning domain. A specific classical planning problem within such a domain is defined by adding an initial state and a goal. The initial state is a conjunction of ground atoms (where the closed-world assumption is assumed to hold). The goal is defined much like the precondition of an action; it is a conjunction of literals (either positive or negative) which may contain

(existentially qualified) variables. A solution to a classical planning problem is a sequence of actions which begin with the initial state, and which entails the goal.

2.2 PDDL

The Planning Domain Definition Language, or PDDL, is the lingua franca in the classical planning community. It was originally developed by the Artificial Intelligence Planning Systems 1998 Competition Committee for use in defining problem domains, and is essentially an expansion of STRIPS. It is not as expressive as some other languages which may be used for planning; however, this is by design, as the planning community wanted the language to be as simple and efficient as possible. Because it is intended for use in an international planning competition, development of the syntax and semantics of PDDL has been largely tied to various competitions. As a result, the development of planners which are able to execute PDDL code to search for plans has been somewhat piecemeal with respect to the formal language specification. That is, particular planners submitted to these competitions may only implement a subset of the features given in the language specification of PDDL.

A particularly striking example of the conflict between the need for PDDL to be simple (so that it may be efficiently parsed) and the desire for it to be expressive may be found in the case of PDDL axioms. Axioms allow for the derivation of some derived predicates, whose value in their current state depends upon some basic predicates (under the closed-world assumption). In [14], it is argued that these axioms not only aid the expressiveness of the language, but also suggests that adding a reasonable implementation which could handle these axioms is beneficial for performance. In many of the domains we discuss in Section 2.4, the ability to use these axioms could have significantly simplified our representation of the problem and domain.

2.3 FF Planner

In order to test the PDDL code we wrote for each of our test domains, we needed to decide on a planner to use. We selected Fast-Forward (FF), because it was simple to compile and run on our systems, and it supported most of the features of PDDL we wanted to be able to take advantage of when encoding the test domains. Most of the information in this section comes from [13].

FF was the most successful automatic planner in the AIPS-2000 planning systems competition. However, the general idea behind FF was not new to the classical planning community - the basic principle is actually the same as that of the Heuristic Search Planner (HSP). FF executes a forward search in the state space, guided by a heuristic which is extracted from the domain description automatically. This function is extracted by relaxing the planning problem; a part of the specification (specifically, the delete lists of all actions) is ignored.

There are a number of details in which FF is different from its predecessor, HSP:

1. FF makes use of a more sophisticated method of heuristic evaluation, which takes into account positive interactions between facts.

2. FF uses a different local search strategy; specifically, it is able to escape plateaus and local minima through the use of systematic search.
3. FF includes a mechanism which identifies those successors of a search node which appear to be (and usually are) most helpful in achieving the goal.

The main difficulty which results from viewing domain independent planning as heuristic search is the automated derivation of the heuristic function. A common approach to this problem (which is adopted here) is to relax the general problem \mathcal{P} into a simpler problem \mathcal{P}' which can be efficiently solved. Given a search state in the original problem, \mathcal{P} , the solution length of the same state \mathcal{P}' may be used to estimate the difficulty of \mathcal{P} . In [11], Bonet et al. propose a way to apply this idea to domain independent planning. In this paper, the high-level problem description is relaxed by “ignoring” delete lists. This means that in the relaxed problem, actions can only add new atoms to state - all negative literals in the effects of actions are ignored, and leave those literals in the state unchanged. As a result, states only grow (in terms of the number of literals present in the state) during execution of a sequence of such relaxed actions, and the problem is solved as soon as each goal literal has been added by some action. This relaxation can be used to derive heuristics that are informative on many benchmark domains in classical planning.

One such heuristic is the length of an optimal relaxed solution. This heuristic is admissible, and could be used to find optimal solution plans through application of A* search; however, computation of the length of an optimal relaxed solution is NP-hard [12]. With this in mind, [11] proposed a way of approximating relaxed solution length from a given search state S , by computing weights over all facts which estimate their distance to S . The key observation which lead to FF’s heuristic method comes from [12]: although computing optimal relaxed solution length is NP-hard, deciding relaxed solvability is actually in P. This means that there must exist polynomial-time decision algorithms, and if such an algorithm constructs a witness, that witness may be used for heuristic evaluation. One algorithmic method which accomplishes this is Graphplan, discussed in [10]. When applied to a solvable relaxed problem, Graphplan finds a solution in polynomial time. Facing a search state S , therefore, FF runs a relaxed version of Graphplan starting at S , and uses the resulting output for heuristic evaluation.

Although these heuristics can be computed in polynomial time, it is still costly to perform heuristic evaluation of states. A logical approach to this issue is to use hill-climbing as the search method, in order to attempt to reach the goal by evaluating as few states as possible. FF makes use of an enforced version of hill-climbing search. When facing a search state S , FF evaluates all direct successors. If none of these successors has a better heuristic value than S , the search then considers the next level (the successors’ successors). Again, if none of the two-step successors looks better than S , FF continues to the next level, and so on. This process ends when a state S' which has a better heuristic evaluation than S is found. The path to S' is then added to the current plan, and the search continues, starting at S' . If a planning problem does not contain dead end situations, this strategy is guaranteed to find a solution.

2.4 Planning Domain Examples

2.4.1 Blocks World

The Blocks World domain is one of the most common benchmark domains in planning as well as in AI; thus, some degree of familiarity with the problem is assumed.

Listing 1: Blocks World Domain Description in PDDL

```

1 (define (domain BLOCKS)
2   (:requirements :strips :typing)
3   (:types block)
4   (:predicates (on ?x - block ?y - block)
5     (ontable ?x - block)
6     (clear ?x - block)
7     (handempty)
8     (holding ?x - block)
9   )
10
11   (:action pick-up
12     :parameters (?x - block)
13     :precondition (and (clear ?x) (ontable ?x) (handempty))
14     :effect (and
15       (not (ontable ?x))
16       (not (clear ?x))
17       (not (handempty))
18       (holding ?x)
19     )
20   )
21
22   (:action put-down
23     :parameters (?x - block)
24     :precondition (holding ?x)
25     :effect (and
26       (not (holding ?x))
27       (clear ?x)
28       (handempty)
29       (ontable ?x)
30     )
31   )
32
33   (:action stack
34     :parameters (?x - block ?y - block)
35     :precondition (and (holding ?x) (clear ?y))
36     :effect (and
37       (not (holding ?x))
38       (not (clear ?y))
39       (clear ?x)
40       (handempty)
41       (on ?x ?y)
42     )
43   )
44
45   (:action unstack
46     :parameters (?x - block ?y - block)
47     :precondition (and (on ?x ?y) (clear ?x) (handempty))
48     :effect (and
49       (holding ?x)
50       (clear ?y)
51       (not (clear ?x))
52       (not (handempty))
53       (not (on ?x ?y))
54     )
55   )
56 )

```

2.4.2 Electrical Circuit




Figure 1: Electrical Circuit Diagram

Listing 2: Electrical Circuit Domain Description in PDDL

```

1  ;; circuit domain
2
3  (define (domain CIRCUIT)
4    (:requirements :typing :conditional-effects)
5    (:types wire gate level)
6    (:predicates
7      (wire-high ?w - wire)
8      (gate-active ?g - gate)
9      (gate-level ?g - gate ?l - level)
10     (and-gate ?g - gate)
11     (or-gate ?g - gate)
12     (inv-gate ?g - gate)
13     (input-to ?w - wire ?g - gate)
14     (output-from ?w - wire ?g - gate)
15   )
16
17   (:action activate-wire
18     :parameters (?w - wire)
19     :vars (?g2 - gate)
20     :precondition (and
21       (not (wire-high ?w))
22       (input-to ?w ?g2)
23       (forall (?g - gate)
24         (not (output-from ?w ?g))
25       )
26     )
27     :effect (and
28       (wire-high ?w)
29     )
30   )
31
32   (:action deactivate-wire
33     :parameters (?w - wire)
34     :vars (?g2 - gate)
35     :precondition (and
36       (wire-high ?w)
37       (input-to ?w ?g2)
38       (forall (?g - gate)
39         (not (output-from ?w ?g))
40       )
41     )

```

```

42     :effect (and
43         (not (wire-high ?w))
44     )
45 )
46 (:action activate-and-gate
47     :parameters (?g - gate)
48     :vars (?w1 - wire ?w2 - wire ?w3 - wire)
49     :precondition (and
50         (and-gate ?g)
51         (not (gate-active ?g))
52         (input-to ?w1 ?g)
53         (input-to ?w2 ?g)
54         (output-from ?w3 ?g)
55         (wire-high ?w1)
56         (wire-high ?w2)
57         (not (= ?w1 ?w2))
58         (not (= ?w1 ?w3))
59         (not (= ?w2 ?w3))
60     )
61     :effect (and
62         (wire-high ?w3)
63         (gate-active ?g)
64     )
65 )
66
67 (:action activate-inv-gate
68     :parameters (?g - gate)
69     :vars (?w1 - wire ?w2 - wire)
70     :precondition (and
71         (inv-gate ?g)
72         (not (gate-active ?g))
73         (input-to ?w1 ?g)
74         (output-from ?w2 ?g)
75         (not (wire-high ?w1))
76     )
77     :effect (and
78         (wire-high ?w2)
79         (gate-active ?g)
80     )
81 )
82
83 (:action activate-or-gate
84     :parameters (?g - gate)
85     :vars (?w1 - wire ?w2 - wire ?w3 - wire)
86     :precondition (and
87         (or-gate ?g)
88         (not (gate-active ?g))
89         (input-to ?w1 ?g)
90         (input-to ?w2 ?g)
91         (output-from ?w3 ?g)
92         (or (wire-high ?w1) (wire-high ?w2))
93     )
94     :effect (and
95         (wire-high ?w3)
96         (gate-active ?g)
97     )
98 )
99 )

```


2.4.3 Lin's Briefcase




Figure 2: Lin's Briefcase Transition Diagram

Listing 3: Lin's Briefcase Domain Description in PDDL

```

1  ;; briefcase domain
2
3  (define (domain BRIEFCASE)
4    (:requirements :typing)
5    (:types latch)
6    (:predicates
7      (open)
8      (latched ?l - latch)
9    )
10
11    (:action flip-open
12      :parameters (?l - latch)
13      :precondition (latched ?l)
14      :effect (not (latched ?l))
15    )
16
17    (:action flip-closed
18      :parameters (?l - latch)
19      :precondition (not (latched ?l))
20      :effect (latched ?l)
21    )
22
23    (:action open
24      :parameters ()
25      :precondition (and
26        (forall (?l - latch)
27          (not (latched ?l))
28        )
29      (not (open))
30    )
31      :effect (open)
32    )
33  )

```

2.4.4 Towers of Hanoi

Listing 4: Towers of Hanoi Domain Description in PDDL

```
1 ;; towers of hanoi
2
3 (define (domain HANOI)
4   (:requirements :typing)
5   (:types disc peg)
6   (:predicates
7     (clear ?x)
8     (on ?x - disc ?y)
9     (larger ?d - disc ?e - disc)
10  )
11
12   (:action stack-d
13     :parameters (?d - disc ?e - disc)
14     :vars (?l)
15     :precondition (and
16       (on ?d ?l)
17       (not (on ?d ?e))
18       (not (= ?d ?e))
19       (not (= ?e ?l))
20       (larger ?e ?d)
21       (clear ?d)
22       (clear ?e)
23     )
24     :effect (and
25       (not (on ?d ?l))
26       (not (clear ?e))
27       (on ?d ?e)
28       (clear ?l)
29     )
30   )
31
32   (:action stack-p
33     :parameters (?d - disc ?p - peg)
34     :vars (?l)
35     :precondition (and
36       (on ?d ?l)
37       (clear ?p)
38       (clear ?d)
39       (not (= ?p ?l))
40     )
41     :effect (and
42       (not (clear ?p))
43       (not (on ?d ?l))
44       (on ?d ?p)
45       (clear ?l)
46     )
47   )
48 )
```

3 The Knowledge Representation and Reasoning Approach

As was stated earlier, in the knowledge representation and reasoning community, the general approach is centered around the use of a portfolio of various knowledge representation languages, among which are the action language \mathcal{AL} , and the logic programming language A-Prolog. In keeping with the notion of programming as the refinement of specification we:

1. Construct a mental model of the transition system described by the domain
2. Describe the transition system via an *action description* in some action language (such as \mathcal{AL})
3. Refine the action description further by translation into A-Prolog

In this section we take a brief look at the languages \mathcal{AL} and A-Prolog, and show their application towards the aforementioned domains.

3.1 The Action Language \mathcal{AL}

Simply put, action languages are a family of declarative languages designed for the purpose of describing the effects of actions. They have a simple syntax and semantics, while remaining powerful enough to represent many complex reasoning domains [3]. Collections of statements are called *action descriptions*, and define transitions whose nodes correspond to *possible states of the world*¹, and whose arcs are labeled by actions. An arc (σ, α, τ) states that if action α occurs in state σ , the resulting state will be τ . In addition to specifying what changes due to the occurrence of a particular action, it is also vital to be able to specify what does not change. This is known as the *frame problem* [4], and its solution lies in an elegant and precise representation of what is called the *inertia axiom*. The beauty and utility of such languages stems from the ability to compactly represent huge transition systems, and in their graceful approach towards solving the frame problem.

Action descriptions in the language of \mathcal{AL} are collections of three statement types: *dynamic causal laws*, *static causal laws*, and *impossibility conditions*. Dynamic causal laws are statements of the form:

$$a \text{ causes } f \text{ if } l_0, \dots, l_n$$

where a is an action, and f , and l_0, \dots, l_n are fluent literals. Laws of this form are read as: “action a causes f to be come true if l_0, \dots, l_n .” The corresponding arc in the transition diagram is shown in Figure 3.




Figure 3: transition defined by a simple causal law

¹A difference we believe to be significant from the conception of a state in the classical view as simply an interpretation of the fluents present in a domain.

Static causal laws have the form:

$$f \text{ if } l_0, \dots, l_n$$

where f , and l_0, \dots, l_n are fluent literals. Such laws are read as: “ f is true whenever l_0, \dots, l_n are true.” Unlike dynamic causal laws, static laws are used to describe the properties of states, as well as the *indirect effects* of actions.

Example 1. Consider the following action description AD_1 :

$$\begin{aligned} & a \text{ causes } f \\ & g \text{ if } f \end{aligned}$$

and let the state, $\sigma = \{\neg f, \neg g\}$. If a is executed in σ , we have the transition depicted in Figure 4.




Figure 4: transition involving both direct and indirect effects

Impossibility conditions have the form:

$$\text{impossible } a \text{ if } l_0, \dots, l_n$$

where as before, a is an action, and l_0, \dots, l_n are fluent literals. Rules such as this are used to state that: “action a may not occur in a state which satisfies l_0, \dots, l_n .” From the view of a transition system, this means that there are no outgoing arcs labeled by a originating from a state which satisfies l_0, \dots, l_n .

The semantics of an action description of \mathcal{AL} is given by its transition diagram. For a more detailed treatment, we refer the reader to [3]. Having described the action language \mathcal{AL} , we now turn our attention to the language of A-Prolog.

3.2 The Language of A-Prolog

The language of A-Prolog is a logic programming language developed by Michael Gelfond and Vladimir Lifschitz in [1]. It has a simple syntax, and is focused on modeling the beliefs of a rational agent. A rather broad community has developed around it, and it has emerged as strong programming paradigm in its own right.

A logic program in the language of A-Prolog is defined as a collection of rules of the form [1, 2]:

$$l_0 \leftarrow l_1, \dots, l_m, \text{not } l_{m+1}, \dots, \text{not } l_n.$$

where $0 \leq m \leq n$, each l_i is a literal, and *not* represents *negation-as-failure* (also termed *default negation*). Rules of this form are read as: “if an agent believes l_1, \dots, l_m to be true and has no reason to believe in l_{m+1}, \dots, l_n , then he believes l_0 to be true.”

The left-hand side of a rule is known as the *head*, and the right-hand side is called the *body*. Rules with empty heads are known as *constraints*, while those with empty bodies are termed *facts*.

The variables of a program Π range over the set of ground terms of its signature σ . Programs which do not contain any variables are said to be *ground*. A rule r which does contain variables may be viewed as a shorthand form for the entire set of its ground instantiations. A *definite rule* is a rule which does not contain any occurrences of default negation, and programs comprised solely of definite rules are called *definite programs*.

Let S be a set of ground literals. The body of a rule is satisfied by S if $\{l_{m+1}, \dots, l_n\} \cap S = \emptyset$ and $\{l_1, \dots, l_m\} \subseteq S$. A rule with a non-empty head is satisfied by S if either its body is not satisfied by S or $l_0 \in S$. A constraint is satisfied by S if its body is not satisfied by S .

Given an arbitrary program Π , and a set of ground literals, S , the *reduct* of Π with respect to S , denoted by Π^S is the definite program obtained from Π by:

1. deleting all rules in Π which not l in the body, where $l \in S$
2. removing all remaining occurrences of not l from the bodies of the remaining rules

A set of ground literals, S , is said to be an answer set of a logic program Π if it satisfies the following:

1. if Π is a definite program then S is a minimal set of literals satisfying all of the rules of Π
2. if Π is not a definite program the S is an answer set of Π if and only if S is an answer set of Π^S

The fact that a program can have multiple (or no) answer sets has given rise to an alternative method of solving problems through logic programming, called *Answer Set Programming* (ASP) [5, 6]. In this approach, we develop logic programs whose answer sets have a one-to-one correspondence with the solutions of the particular problem being modeled. Typically an ASP program consists of:

- rules for the enumeration of possible solutions to a problem as *candidate answer sets*
- constraints which remove candidates that do not correspond to solutions of the problem

In the context of planning, the “planning module” will serve to generate possible sequences of while the “goal specification” will prune those which fail to satisfy our goal.

3.3 Domains

In this section we examine a number of planning and apply the knowledge representation and reasoning approach towards their resolution. All of the listings presented in this paper are in the input language of the clingo answer set programming system [7].

3.3.1 Lin's Briefcase

Consider a domain consisting of a briefcase with two latches. Flipping a closed latch causes it to be open, while flipping an open latch causes it to become closed. How do we represent this domain using the reasoning about actions approach?

We begin by constructing a mental model of the system, which involves the transition system shown in Figure 5.




Figure 5: transition system for the Lin's Briefcase domain

Once we have the transition fixed, we represent in \mathcal{AL} as follows:

$\text{flip}(l_1)$ causes $\text{open}(l_1)$ if $\neg \text{open}(l_1)$
 $\text{flip}(l_1)$ causes $\neg \text{open}(l_1)$ if $\text{open}(l_1)$
 $\text{flip}(l_2)$ causes $\text{open}(l_2)$ if $\neg \text{open}(l_2)$
 $\text{flip}(l_2)$ causes $\neg \text{open}(l_2)$ if $\text{open}(l_2)$
 $\text{open}(b)$ if $\text{open}(l_1), \text{open}(l_2)$
 $\neg \text{open}(b)$ if $\neg \text{open}(l_1)$
 $\neg \text{open}(b)$ if $\neg \text{open}(l_2)$

With our domain description in place, we translate the above laws and add some additional domain related information to obtain the following A-Prolog program:

Listing 5: Lin's Briefcase Domain Description in A-Prolog

```

1  % =====
2  % OBJECTS
3  % =====
4
5  #const n = 2.
6
7  time(0..n).
8
9  latch(1).
10 latch(2).
11
12 % =====
13 % FLUENTS
14 % =====
15
16 fluent(open(Latch)) :-
17     latch(Latch).
18

```

```

19 fluent(opened).
20
21 % =====
22 % ACTIONS
23 % =====
24
25 action(flip(Latch)) :-
26     latch(Latch).
27
28 % =====
29 % CAUSAL LAWS
30 % =====
31
32 holds(open(Latch), T + 1) :-
33     occurs(flip(Latch), T),
34     -holds(open(Latch), T),
35     latch(Latch),
36     time(T).
37
38 -holds(open(Latch), T + 1) :-
39     occurs(flip(Latch), T),
40     holds(open(Latch), T),
41     latch(Latch),
42     time(T).
43
44 -holds(opened, T) :-
45     -holds(open(Latch), T),
46     latch(Latch),
47     time(T).
48
49 holds(opened, T) :-
50     not -holds(opened, T),
51     time(T).
52
53 % =====
54 % INERTIA AXIOMS
55 % =====
56
57 holds(Fluent, T + 1) :-
58     holds(Fluent, T),
59     not -holds(Fluent, T + 1),
60     fluent(Fluent),
61     time(T).
62
63 -holds(Fluent, T + 1) :-
64     -holds(Fluent, T),
65     not holds(Fluent, T + 1),
66     fluent(Fluent),
67     time(T).
68
69 % =====
70 % INITIAL STATE
71 % =====
72
73 -holds(open(Latch), 0) :-
74     latch(Latch).
75
76 % =====

```

```

77 % GOAL SPECIFICATION
78 % =====
79
80 goal(T) :-
81     holds(opened, T),
82     time(T).
83
84 done :-
85     goal(T).
86
87 :- not done.
88
89 % =====
90 % PLANNING MODULE
91 % =====
92
93 l{occurs(Action,T) : action(Action)}1 :-
94     not goal(T),
95     time(T).
96
97 % =====
98 % SOLVER DIRECTIVES
99 % =====
100
101 #hide.
102 #show occurs(_,_).

```

Utilizing the clingo answer set solver we obtain the following two plans (this is due to the fact that the planning module is written in such a way as to only look for sequential plans, parallel plans may be supported through a minor modification):

```

Answer: 1
occurs(flip(2),1) occurs(flip(1),0)
Answer: 2
occurs(flip(1),1) occurs(flip(2),0)
SATISFIABLE

```

```

Models      : 2
Time        : 0.000
  Prepare   : 0.000
  Prepro.   : 0.000
  Solving   : 0.000

```

3.3.2 The Blocks World

The next domain we examine is the Blocks Worlds. In this domain an agent seeks to build a tower out of a set blocks. This domain makes full use of the full range of causal laws provided for in \mathcal{AL} . For the sake of space, the \mathcal{AL} description of the system has not been included, however, a reading of the relevant causal laws in the A-Prolog representation will show that each type of law is used and follows the same representation scheme we have seen previously with Lin's Briefcase.

Listing 6: Blocks World Domain Representation in A-Prolog

```

1  % =====
2  % OBJECTS
3  % =====
4
5  #const n = 3.
6
7  time(0..n).
8
9  block(a).
10 block(b).
11 block(c).
12 block(d).
13
14 table(t).
15
16 location(X) :- block(X).
17 location(X) :- table(X).
18
19 % =====
20 % FLUENTS
21 % =====
22
23 fluent(on(Block,Location)) :-
24     Block != Location,
25     block(Block),
26     location(Location).
27
28 fluent(clear(Location)) :-
29     location(Location).
30
31 % =====
32 % ACTIONS
33 % =====
34
35 action(move(Block,Location)) :-
36     Block != Location,
37     block(Block),
38     location(Location).
39
40 % =====
41 % CAUSAL LAWS
42 % =====
43
44 holds(on(Block,Location), T + 1) :-
45     occurs(move(Block,Location), T),
46     block(Block),
47     location(L),
48     time(T).
49
50 :- occurs(move(Block,Location), T),
51     not holds(clear(Location), T),
52     block(Block),
53     location(Location),
54     time(T).
55
56 :- occurs(move(Block,Location), T),
57     not holds(clear(Block), T),

```

```

58     block(Block),
59     location(Location),
60     time(T).
61
62 -holds(on(Block,Location2), T) :-
63     holds(on(Block,Location1), T),
64     Location1 != Location2,
65     block(Block),
66     location(Location1),
67     location(Location2),
68     time(T).
69
70 -holds(clear(Block1), T) :-
71     holds(on(Block2,Block1), T),
72     block(Block1),
73     block(Block2),
74     time(T).
75
76 holds(clear(Table), T) :-
77     table(Table),
78     time(T).
79
80 holds(clear(Block), T) :-
81     not -holds(clear(Block), T),
82     block(Block),
83     time(T).
84
85 % =====
86 % INERTIA AXIOMS
87 % =====
88
89 holds(Fluent, T + 1) :-
90     holds(Fluent, T),
91     not -holds(Fluent, T + 1),
92     fluent(Fluent),
93     time(T).
94
95 -holds(Fluent, T + 1) :-
96     -holds(Fluent, T),
97     not holds(Fluent, T + 1),
98     fluent(Fluent),
99     time(T).
100
101 % =====
102 % INITIAL STATE
103 % =====
104
105 holds(on(a,t), 0).
106 holds(on(b,t), 0).
107 holds(on(c,t), 0).
108 holds(on(d,t), 0).
109
110 % =====
111 % GOAL SPECIFICATION
112 % =====
113
114 goal(T) :-
115     holds(on(d,c), T),

```

```

116         holds(on(c,b), T),
117         holds(on(b,a), T),
118         holds(on(a,t), T),
119         time(T).
120
121 done :-
122     goal(T).
123
124 :- not done.
125
126 % =====
127 % PLANNING MODULE
128 % =====
129
130 l{occurs(Action,T) : action(Action)}1 :-
131     not goal(T),
132     time(T).
133
134 % =====
135 % SOLVER DIRECTIVES
136 % =====
137
138 #hide.
139 #show occurs(_,_).

```

Once again, using the clingo answer set solver we obtain the following single plan:

```

Answer: 1
occurs(move(d,c),2) occurs(move(c,b),1) occurs(move(b,a),0)
SATISFIABLE

```

```

Models      : 1
Time        : 0.000
  Prepare   : 0.000
  Prepro.   : 0.000
  Solving   : 0.000

```

3.3.3 Electrical Circuit

We now turn our attention towards the representation of an electrical circuit domain. Such a domain is of interest due to the fact that there is essentially only a single action that an agent may perform, namely, the application of some input signals to the circuit. These signals must then be propagated throughout the entire circuit to obtain some output value.

Consider the electrical circuit shown in Figure 6. Once a signal has been applied to inputs 1,2,3, it must be propagated to the next level of the circuit within the same state. Static causal laws shine here.

Listing 7: Electrical Circuit Domain Representation in A-Prolog

```

1 % =====
2 % OBJECTS
3 % =====

```




Figure 6: a simple electrical circuit

```

4
5 time(0).
6
7 % =====
8 % CIRCUIT STRUCTURE
9 % =====
10
11 wire(1..6).
12 gate(1..3).
13 and(1).
14 inverter(2).
15 or(3).
16
17 % and gate has inputs 1 and 2
18
19 input_to(1,1..2).
20
21 % inverter has input 3
22
23 input_to(2,3).
24
25 % or gate has inputs 4 and 5
26
27 input_to(3,4..5).
28
29 % and gate has output 4
30
31 output_from(1,4).
32
33 % inverter has output 5
34
35 output_from(2,5).
36
37 % or gate has output 6
38
39 output_from(3,6).
40
41 % =====
42 % GENERAL DESCRIPTION OF GATES
43 % =====
44
45 h(val(OUTPUT_WIRE, 1), T) :-
46     output_from(GATE, OUTPUT_WIRE),
47     and(GATE),
48     input_to(GATE, INPUT1),

```

```

49     input_to(GATE, INPUT2),
50     INPUT1 != INPUT2,
51     h(val(INPUT1, 1), T),
52     h(val(INPUT2, 1), T),
53     time(T).
54
55 h(val(OUTPUT_WIRE, 0), T) :-
56     output_from(GATE, OUTPUT_WIRE),
57     and(GATE),
58     input_to(GATE, INPUT),
59     h(val(INPUT, 0), T),
60     time(T).
61
62 h(val(OUTPUT_WIRE, 0), T) :-
63     output_from(GATE, OUTPUT_WIRE),
64     or(GATE),
65     input_to(GATE, INPUT1),
66     input_to(GATE, INPUT2),
67     INPUT1 != INPUT2,
68     h(val(INPUT1, 0), T),
69     h(val(INPUT2, 0), T),
70     time(T).
71
72 h(val(OUTPUT_WIRE, 1), T) :-
73     output_from(GATE, OUTPUT_WIRE),
74     or(GATE),
75     input_to(GATE, INPUT),
76     h(val(INPUT, 1), T),
77     time(T).
78
79 h(val(OUTPUT_WIRE, 1), T) :-
80     output_from(GATE, OUTPUT_WIRE),
81     inverter(GATE),
82     input_to(GATE, INPUT),
83     h(val(INPUT, 0), T),
84     time(T).
85
86 h(val(OUTPUT_WIRE, 0), T) :-
87     output_from(GATE, OUTPUT_WIRE),
88     inverter(GATE),
89     input_to(GATE, INPUT),
90     h(val(INPUT, 1), T),
91     time(T).
92
93 % =====
94 % INITIAL STATE
95 % =====
96
97 h(val(1, 1), 0).
98 h(val(2, 0), 0).
99 h(val(3, 1), 0).
100
101 % =====
102 % SOLVER DIRECTIVES
103 % =====
104
105 #hide.
106 #show h(_, _).

```

Once again using clingo, we can see how the signal propagates across the various levels of the circuit:

```
Answer: 1
h(val(1,1),0) h(val(2,0),0) h(val(3,1),0) h(val(4,0),0) h(val(5,0),0) h(val(6,0),0)
SATISFIABLE
```

```
Models      : 1
Time        : 0.000
  Prepare   : 0.000
  Prepro.   : 0.000
  Solving   : 0.000
```

3.3.4 Towers of Hanoi

The last domain we consider is the Towers of Hanoi domain. This domain was chosen due to the similarity of the PDDL and A-Prolog representations. The PDDL representation allows one to simulate in convenient “lossless” way the information stored in a static causal law concerning whether a disc or peg becomes clear. It is also an interesting planning problem in its own right.

Listing 8: Towers of Hanoi Domain Representation in A-Prolog

```
1 % =====
2 % OBJECTS
3 % =====
4
5 #const n = 7.
6
7 time(0..n).
8
9 disc(1).
10 disc(2).
11 disc(3).
12
13 peg(p1).
14 peg(p2).
15 peg(p3).
16
17 location(X) :- peg(X).
18 location(X) :- disc(X).
19
20 % =====
21 % FLUENTS
22 % =====
23
24 fluent(on(Disc,Location)) :-
25     disc(Disc),
26     location(Location),
27     Disc != Location.
28
29 fluent(clear(Location)) :-
30     location(Location).
31
32 % =====
```

```

33 % ACTIONS
34 % =====
35
36 action(move(Disc,Source,Destination)) :-
37     disc(Disc),
38     location(Source),
39     location(Destination),
40     Source != Destination.
41
42 % =====
43 % CAUSAL LAWS
44 % =====
45
46 holds(on(Disc,Destination), T + 1) :-
47     occurs(move(Disc,Source,Destination), T),
48     disc(Disc),
49     location(Source),
50     location(Destination),
51     time(T).
52
53 -holds(on(Disc,Source), T + 1) :-
54     occurs(move(Disc,Source,Destination), T),
55     disc(Disc),
56     location(Source),
57     location(Destination),
58     time(T).
59
60 :- occurs(move(Disc,Source,Destination), T),
61     not holds(on(Disc,Source), T),
62     disc(Disc),
63     location(Source),
64     location(Destination),
65     time(T).
66
67 :- occurs(move(Disc,Source,Destination), T),
68     not holds(clear(Disc), T),
69     disc(Disc),
70     location(Source),
71     location(Destination),
72     time(T).
73
74 :- occurs(move(Disc,Source,Destination), T),
75     not holds(clear(Destination), T),
76     disc(Disc),
77     location(Source),
78     location(Destination),
79     time(T).
80
81 :- occurs(move(Disc,Source,Destination), T),
82     disc(Destination),
83     Destination < Disc,
84     disc(Disc),
85     location(Source),
86     location(Destination),
87     time(T).
88
89 -holds(clear(Location), T) :-
90     holds(on(Disc,Location), T),

```

```

91     location(Location),
92     disc(Disc),
93     time(T).
94
95 holds(clear(Location), T) :-
96     not -holds(clear(Location), T),
97     location(Location),
98     time(T).
99
100 % =====
101 % INERTIA AXIOMS
102 % =====
103
104 holds(Fluent, T + 1) :-
105     holds(Fluent, T),
106     not -holds(Fluent, T + 1),
107     fluent(Fluent),
108     time(T).
109
110 -holds(Fluent, T + 1) :-
111     -holds(Fluent, T),
112     not holds(Fluent, T + 1),
113     fluent(Fluent),
114     time(T).
115
116 % =====
117 % INITIAL STATE
118 % =====
119
120 holds(on(1,2), 0).
121 holds(on(2,3), 0).
122 holds(on(3,p1), 0).
123
124 % =====
125 % GOAL SPECIFICATION
126 % =====
127
128 goal(T) :-
129     holds(on(1,2), T),
130     holds(on(2,3), T),
131     holds(on(3,p3), T),
132     time(T).
133
134 done :-
135     goal(T).
136
137 :- not done.
138
139 % =====
140 % PLANNING MODULE
141 % =====
142
143 l{occurs(Action,T) : action(Action)}1 :-
144     not goal(T),
145     time(T).
146
147 % =====
148 % SOLVER DIRECTIVES

```



```

149 % =====
150
151 #hide.
152 #show occurs(_,-).

```

Once again clingo gives us a plan for moving the disc to the destination peg:

```

Answer: 1
occurs(move(1,p1,2),6) occurs(move(2,p2,3),5) occurs(move(1,2,p1),4)
occurs(move(3,p1,p3),3) occurs(move(1,p3,2),2) occurs(move(2,3,p2),1)
occurs(move(1,2,p3),0)
SATISFIABLE

```

```

Models      : 1+
Time        : 0.030
  Prepare   : 0.020
  Prepro.   : 0.010
  Solving   : 0.000

```

4 Conclusions

In keeping with Dijkstra’s view of programming as the refinement of specification, we believe the approach taken by the knowledge representation and reasoning community to be more expressive. It has been shown to be capable of handling extremely robust domains such as the reaction control system of the space shuttle, [9], and has been applied towards the task of *conformant planning* as well [8]. The central concept of reducing the task of planning to the computation of the answer sets of a logic program separates the quality and type of plans generated from the solver being used for their computation. The two become independent, and any improvements to the solving algorithm, have no effect on which plans are returned (as plans are simply the answer set of program). This is not the case in classical planning. In the classical planning approach the type and quality of plans produced are highly dependent on the type of solver being used. This makes it difficult to say what the actual *meaning* of a PDDL domain representation actually is. It also makes it difficult to make general claims of the properties of PDDL domain description.

The two communities also use different notions of what a state is. In the classical planning community, a state is simply an interpretation of the fluents. In the knowledge representation community, states are interpretations that are closed with respect to all of the state constraints present in an action description. This has some profound consequences, particularly when coupled with many classical planners not supporting a feature known as domain axioms. Consider again Lin’s Briefcase domain, and suppose we are in the state $\{\text{open}(l_1), \text{open}(l_2), \text{open}(b)\}$. Now suppose that we close one of the latches, l_1 . Through the use of static causal laws, under the knowledge representation approach the successor state will be $\{\neg \text{open}(l_1), \text{open}(l_2), \neg \text{open}(b)\}$. Under the classical planning approach however, we would have this $\{\neg \text{open}(l_1), \text{open}(l_2), \text{open}(b)\}$ as a successor state, which violates our domain specification. Similar issues occur with the Electrical Circuit domain, where the interior elements have essentially undefined values until all of the requisite PDDL “propagation” actions have fired.

This lack of expressiveness on the part of PDDL is an intentional decision given that its express purpose was to serve as a vehicle for the planning competitions, and not as a knowledge representation language. We regret not having had the time to do some thorough benchmarking of some complex domains pitting the two approaches against each other. For certain domains, we believe that classical planners may have an edge, particularly those involving a considerable amount of numeric computation. For others however, we believe that given its impressive performance with regards to the shuttle [9], and other complex domains, A-Prolog and the knowledge representation approach will prove to be quite competitive.

References

- [1] Michael Gelfond and Vladimir Lifschitz. The Stable Model Semantics for Logic Programming. In Robert A. Kowalski and Kenneth Bowen, editors, *Proceedings of the Fifth International Conference on Logic Programming*, pages 1070-1080, Cambridge, Massachusetts, 1988. The MIT Press.
- [2] Chitta Baral. *Knowledge Representation, Reasoning, and Declarative Problem Solving*. Cambridge University Press, January 2003.
- [3] Marcello Balduccini. *Answer Set Based Design of Highly Autonomous, Rational Agents*. PhD thesis, Texas Tech University, December 2005.
- [4] Murray Shanahan. *Solving the frame problem: A mathematical investigation of the commonsense law of inertia*. MIT Press, 1997.
- [5] Victor Marek and Mirek Truszczyński. Stable models and an alternative logic programming paradigm. *The Logic Programming Paradigm*, pp. 375–398, Springer, 1999.
- [6] Ilkka Niemela. Logic programming with stable model semantics as a constraint programming paradigm. *AMAI*, 25(3,4), 1999.
- [7] Martin Gebser, Roland Kaminski, et. al. *A User’s Guide to gringo, clasp, clingo, and iclingo*. 2010.
- [8] Michael Gelfond and Ricardo Morales. Encoding Conformant Planning in A-Prolog. In *DRT’04*, 2004.
- [9] Marcello Balduccini, Michael Gelfond, Monica Nogueira, and Richard Watson. Planning with the USA-Advisor. In *3rd International NASA Workshop on Planning and Scheduling for Space*, Sep 2002.
- [10] A. Blum and M. Furst. Fast planning through planning graph analysis. *Artificial Intelligence*, 90(1-2):279–298, 1997.
- [11] B. Bonet, G. Loernics, and H. Geffner. A robust and fast action selection mechanism for planning. In *AAAI-97*, pages 714–719. MIT Press, 1997.
- [12] T. Bylander. The computational complexity of propositional strips planning. *Artificial Intelligence*, 69(1-2):165–204, 1994.
- [13] J. Hoffmann. Ff: The fast-forward planning system. *AI Magazine*, 22(3):57–62, 2001.
- [14] S. Thiébaux, J. Hoffmann, and B. Nebel. In defense of pddl axioms. *Artificial Intelligence*, 18:961–968, 2003.

5 Roles in the Project

Name	Contributions
Gregory Gelfond	<ul style="list-style-type: none">• Read cited papers• Wrote A-Prolog problem and domain encodings• Compiled project presentation• Wrote report introduction• Wrote knowledge representation and reasoning section of the report• Contributed to comparison and analysis of methodologies
Glen Hunt	<ul style="list-style-type: none">• Read cited papers• Contributed classical planning information to the project presentation• Wrote PDDL problem and domain encodings• Wrote classical planning section of the report• Contributed to comparison and analysis of methodologies