

1 Classical Planning Overview

1.1 Background

Planning, in general, consists of computing a sequence of actions which, starting in an initial state or states, will achieve a set of goal states. Classical planning, then, is a constrained planning task, which is fully observable, deterministic, finite, static (i.e. the world does not change unless the agent acts), and discrete (i.e. time, actions, objects and effects are not continuous). Nonclassical planning, by contrast, is used to describe planning in partially observable or stochastic environments.

1.1.1 Difficulties of Classical Planning

Typical problem-solving agents which make use of standard search algorithms (i.e. depth-first, breadth-first, A^* , etc) encounter several difficulties when solving classical planning problems; each of these must be kept in mind when designing a planning agent. First of all, the agent may be overwhelmed by “irrelevant” actions (that is, actions which do not necessarily bring the agent closer to achieving a goal). There are multiple ways within the classical planning community of discouraging a planning agent from considering irrelevant actions; the use of regression search, for example, is one way to attempt to handle this problem.

Another difficulty in classical planning is finding a good heuristic function. Because a problem-solving agent would see the goal state only as a “black box” which is either true or false for a given state, it would require a new heuristic for each specific problem it encounters. A planning agent, however, may have access to a representation of the goal state as a conjunction of subgoals, in which case it can use a single domain-independent heuristic; namely, the number of unsatisfied conjuncts.

Similarly, a problem-solving agent might be inefficient because of its inability to take advantage of problem decomposition. A planning agent which views a state as a conjunction of literals, however, might be able to decompose the goal into a conjunction of subgoals, and then attempt to find a plan to achieve each subgoal. There are tradeoffs to this approach, however. Even if a planning agent can work on subgoals independently, some additional work may be required to combine the resultant subplans. Additionally, for some domains, one subgoal may undo another.

1.1.2 Representing Classical Planning Problems

Classical planning problems are represented using three formalisms: states, goals and actions. These constructs form the basis of the languages used to encode planning problems in such a way that planners can understand them and operate on them to search for plans.

States in classical planning problems are represented by a conjunction of positive, ground, first-order propositional literals. Classical planning adopts the closed-world assumption, which states that any literals which are not explicitly enumerated as part of the state are false. This state representation

may be manipulated by logical inference (when viewed as a conjunction of fluents), or with set operations (when viewed as a set of fluents).

Actions may be represented by an action schema. This schema consists of the action name, a list of all variables used in the schema, a precondition, and an effect. Both the precondition and effect of the action are conjunctions of literals, which may be either positive or negative. The precondition gives a definition of states in which the action can be executed; the effect gives the result of executing the action. More formally, an action a may be executed in some state s if s entails the precondition of a . If an action schema contains variables, there may be multiple states (ground instantiations of the variables in the action) which satisfy the preconditions of the action.

The result of executing some action a in state s is given by state s' . s' is first populated with the set of fluents in s , then all fluents which appear as negative literals in the effects of a are removed, and finally the fluents which appear as positive literals in the effects of a are added.

A set of these action schemas gives the definition of a planning domain. A specific classical planning problem within such a domain is defined by adding an initial state and a goal. The initial state is a conjunction of ground atoms (where the closed-world assumption is assumed to hold). The goal is defined much like the precondition of an action; it is a conjunction of literals (either positive or negative) which may contain (existentially qualified) variables. A solution to a classical planning problem is a sequence of actions which begin with the initial state, and which entails the goal.

1.2 PDDL

The Planning Domain Definition Language, or PDDL, is the lingua franca in the classical planning community. It was originally developed by the Artificial Intelligence Planning Systems 1998 Competition Committee for use in defining problem domains, and is essentially an expansion of STRIPS. It is not as expressive as some other languages which may be used for planning; however, this is by design, as the planning community wanted the language to be as simple and efficient as possible. Because it is intended for use in an international planning competition, development of the syntax and semantics of PDDL has been largely tied to various competitions. As a result, the development of planners which are able to execute PDDL code to search for plans has been somewhat piecemeal with respect to the formal language specification. That is, particular planners submitted to these competitions may only implement a subset of the features given in the language specification of PDDL.

A particularly striking example of the conflict between the need for PDDL to be simple (so that it may be efficiently parsed) and the desire for it to be expressive may be found in the case of PDDL axioms. Axioms allow for the derivation of some derived predicates, whose value in their current state depends upon some basic predicates (under the closed-world assumption). In [5], it is argued that these axioms not only aid the expressiveness of the language, but also suggests that adding a reasonable implementation which could handle these axioms is beneficial for performance. In many of the domains we discuss in Section 1.4, the ability to use these axioms could have significantly simplified our representation of the problem and domain.

1.3 FF Planner

In order to test the PDDL code we wrote for each of our test domains, we needed to decide on a planner to use. We selected Fast-Forward (FF), because it was simple to compile and run on our systems, and it supported most of the features of PDDL we wanted to be able to take advantage of when encoding the test domains. Most of the information in this section comes from [4].

FF was the most successful automatic planner in the AIPS-2000 planning systems competition. However, the general idea behind FF was not new to the classical planning community - the basic principle is actually the same as that of the Heuristic Search Planner (HSP). FF executes a forward search in the state space, guided by a heuristic which is extracted from the domain description automatically. This function is extracted by relaxing the planning problem; a part of the specification (specifically, the delete lists of all actions) is ignored.

There are a number of details in which FF is different from its predecessor, HSP:

1. FF makes use of a more sophisticated method of heuristic evaluation, which takes into account positive interactions between facts.
2. FF uses a different local search strategy; specifically, it is able to escape plateaus and local minima through the use of systematic search.
3. FF includes a mechanism which identifies those successors of a search node which appear to be (and usually are) most helpful in achieving the goal.

The main difficulty which results from viewing domain independent planning as heuristic search is the automated derivation of the heuristic function. A common approach to this problem (which is adopted here) is to relax the general problem \mathcal{P} into a simpler problem \mathcal{P}' which can be efficiently solved. Given a search state in the original problem, \mathcal{P} , the solution length of the same state \mathcal{P}' may be used to estimate the difficulty of \mathcal{P} . In [2], Bonet et al. propose a way to apply this idea to domain independent planning. In this paper, the high-level problem description is relaxed by “ignoring” delete lists. This means that in the relaxed problem, actions can only add new atoms to state - all negative literals in the effects of actions are ignored, and leave those literals in the state unchanged. As a result, states only grow (in terms of the number of literals present in the state) during execution of a sequence of such relaxed actions, and the problem is solved as soon as each goal literal has been added by some action. This relaxation can be used to derive heuristics that are informative on many benchmark domains in classical planning.

One such heuristic is the length of an optimal relaxed solution. This heuristic is admissible, and could be used to find optimal solution plans through application of A^* search; however, computation of the length of an optimal relaxed solution is NP-hard [3]. With this in mind, [2] proposed a way of approximating relaxed solution length from a given search state \mathcal{S} , by computing weights over all facts which estimate their distance to \mathcal{S} . The key observation which lead to FF’s heuristic method comes from [3]: although computing optimal relaxed solution length is NP-hard, deciding relaxed solvability is actually in P. This means that there must exist polynomial-time decision algorithms, and if such an algorithm constructs a witness, that witness may be used for heuristic evaluation.

One algorithmic method which accomplishes this is Graphplan, discussed in [1]. When applied to a solvable relaxed problem, Graphplan finds a solution in polynomial time. Facing a search state \mathcal{S} , therefore, FF runs a relaxed version of Graphplan starting at \mathcal{S} , and uses the resulting output for heuristic evaluation.

Although these heuristics can be computed in polynomial time, it is still costly to perform heuristic evaluation of states. A logical approach to this issue is to use hill-climbing as the search method, in order to attempt to reach the goal by evaluating as few states as possible. FF makes use of an enforced version of hill-climbing search. When facing a search state \mathcal{S} , FF evaluates all direct successors. If none of these successors has a better heuristic value than \mathcal{S} , the search then considers the next level (the successors' successors). Again, if none of the two-step successors looks better than \mathcal{S} , FF continues to the next level, and so on. This process ends when a state \mathcal{S}' which has a better heuristic evaluation than \mathcal{S} is found. The path to \mathcal{S}' is then added to the current plan, and the search continues, starting at \mathcal{S}' . If a planning problem does not contain dead end situations, this strategy is guaranteed to find a solution.

1.4 Planning Domain Examples

1.4.1 Blocks World

The Blocks World domain is one of the most common benchmark domains in planning as well as in AI; as a result, some degree of familiarity with the problem is assumed.

Listing 1: Blocks World Domain Description in PDDL

```

1 (define (domain BLOCKS)
2   (:requirements :strips :typing)
3   (:types block)
4   (:predicates (on ?x - block ?y - block)
5     (ontable ?x - block)
6     (clear ?x - block)
7     (handempty)
8     (holding ?x - block)
9   )
10
11   (:action pick-up
12     :parameters (?x - block)
13     :precondition (and (clear ?x) (ontable ?x) (handempty))
14     :effect (and
15       (not (ontable ?x))
16       (not (clear ?x))
17       (not (handempty))
18       (holding ?x)
19     )
20   )
21
22   (:action put-down
23     :parameters (?x - block)

```

```

24     :precondition (holding ?x)
25     :effect (and
26         (not (holding ?x))
27         (clear ?x)
28         (handempty)
29         (ontable ?x)
30     )
31 )
32
33 (:action stack
34   :parameters (?x - block ?y - block)
35   :precondition (and (holding ?x) (clear ?y))
36   :effect (and
37       (not (holding ?x))
38       (not (clear ?y))
39       (clear ?x)
40       (handempty)
41       (on ?x ?y)
42   )
43 )
44
45 (:action unstack
46   :parameters (?x - block ?y - block)
47   :precondition (and (on ?x ?y) (clear ?x) (handempty))
48   :effect (and
49       (holding ?x)
50       (clear ?y)
51       (not (clear ?x))
52       (not (handempty))
53       (not (on ?x ?y))
54   )
55 )
56 )

```

1.4.2 Electrical Circuit

Listing 2: Electrical Circuit Domain Description in PDDL

```

1  ;; circuit domain
2
3  (define (domain CIRCUIT)
4    (:requirements :typing :conditional-effects)
5    (:types wire gate level)
6    (:predicates
7      (wire-high ?w - wire)
8      (gate-active ?g - gate)
9      (gate-level ?g - gate ?l - level)
10     (and-gate ?g - gate)
11     (or-gate ?g - gate)
12     (inv-gate ?g - gate)
13     (input-to ?w - wire ?g - gate)

```

```

14         (output-from ?w - wire ?g - gate)
15     )
16
17 (:action activate-wire
18   :parameters (?w - wire)
19   :vars (?g2 - gate)
20   :precondition (and
21     (not (wire-high ?w))
22     (input-to ?w ?g2)
23     (forall (?g - gate)
24       (not (output-from ?w ?g))
25     )
26   )
27   :effect (and
28     (wire-high ?w)
29   )
30 )
31
32 (:action deactivate-wire
33   :parameters (?w - wire)
34   :vars (?g2 - gate)
35   :precondition (and
36     (wire-high ?w)
37     (input-to ?w ?g2)
38     (forall (?g - gate)
39       (not (output-from ?w ?g))
40     )
41   )
42   :effect (and
43     (not (wire-high ?w))
44   )
45 )
46 (:action activate-and-gate
47   :parameters (?g - gate)
48   :vars (?w1 - wire ?w2 - wire ?w3 - wire)
49   :precondition (and
50     (and-gate ?g)
51     (not (gate-active ?g))
52     (input-to ?w1 ?g)
53     (input-to ?w2 ?g)
54     (output-from ?w3 ?g)
55     (wire-high ?w1)
56     (wire-high ?w2)
57     (not (= ?w1 ?w2))
58     (not (= ?w1 ?w3))
59     (not (= ?w2 ?w3))
60   )
61   :effect (and
62     (wire-high ?w3)
63     (gate-active ?g)
64   )
65 )
66

```

```

67 (:action activate-inv-gate
68   :parameters (?g - gate)
69   :vars (?w1 - wire ?w2 - wire)
70   :precondition (and
71     (inv-gate ?g)
72     (not (gate-active ?g))
73     (input-to ?w1 ?g)
74     (output-from ?w2 ?g)
75     (not (wire-high ?w1))
76   )
77   :effect (and
78     (wire-high ?w2)
79     (gate-active ?g)
80   )
81 )
82
83 (:action activate-or-gate
84   :parameters (?g - gate)
85   :vars (?w1 - wire ?w2 - wire ?w3 - wire)
86   :precondition (and
87     (or-gate ?g)
88     (not (gate-active ?g))
89     (input-to ?w1 ?g)
90     (input-to ?w2 ?g)
91     (output-from ?w3 ?g)
92     (or (wire-high ?w1) (wire-high ?w2))
93   )
94   :effect (and
95     (wire-high ?w3)
96     (gate-active ?g)
97   )
98 )
99 )

```

1.4.3 Lin's Briefcase

Listing 3: Lin's Briefcase Domain Description in PDDL

```

1  ;; briefcase domain
2
3  (define (domain BRIEFCASE)
4    (:requirements :typing)
5    (:types latch)
6    (:predicates
7      (open)
8      (latched ?l - latch)
9    )
10
11    (:action flip-open
12      :parameters (?l - latch)
13      :precondition (latched ?l)

```

```

14     :effect (not (latched ?l))
15   )
16
17   (:action flip-closed
18     :parameters (?l - latch)
19     :precondition (not (latched ?l))
20     :effect (latched ?l)
21   )
22
23   (:action open
24     :parameters ()
25     :precondition (and
26       (forall (?l - latch)
27         (not (latched ?l))
28       )
29     (not (open)))
30   )
31   :effect (open)
32 )
33 )

```

1.4.4 Towers of Hanoi

Listing 4: Towers of Hanoi Domain Description in PDDL

```

1  ;; towers of hanoi
2
3  (define (domain HANOI)
4    (:requirements :typing)
5    (:types disc peg)
6    (:predicates
7      (clear ?x)
8      (on ?x - disc ?y)
9      (larger ?d - disc ?e - disc)
10   )
11
12   (:action stack-d
13     :parameters (?d - disc ?e - disc)
14     :vars (?l)
15     :precondition (and
16       (on ?d ?l)
17       (not (on ?d ?e))
18       (not (= ?d ?e))
19       (not (= ?e ?l))
20       (larger ?e ?d)
21       (clear ?d)
22       (clear ?e)
23     )
24     :effect (and
25       (not (on ?d ?l))
26       (not (clear ?e))

```



```

27         (on ?d ?e)
28         (clear ?l)
29     )
30 )
31
32 (:action stack-p
33   :parameters (?d - disc ?p - peg)
34   :vars (?l)
35   :precondition (and
36     (on ?d ?l)
37     (clear ?p)
38     (clear ?d)
39     (not (= ?p ?l))
40   )
41   :effect (and
42     (not (clear ?p))
43     (not (on ?d ?l))
44     (on ?d ?p)
45     (clear ?l)
46   )
47 )
48 )

```

References

- [1] A. Blum and M. Furst. Fast planning through planning graph analysis. *Artificial Intelligence*, 90(1-2):279–298, 1997.
- [2] B. Bonet, G. Loernics, and H. Geffner. A robust and fast action selection mechanism for planning. In *AAAI-97*, pages 714–719. MIT Press, 1997.
- [3] T. Bylander. The computational complexity of propositional strips planning. *Artificial Intelligence*, 69(1-2):165–204, 1994.
- [4] J. Hoffmann. Ff: The fast-forward planning system. *AI Magazine*, 22(3):57–62, 2001.
- [5] S. Thiébaux, J. Hoffmann, and B. Nebel. In defense of pddl axioms. *Artificial Intelligence*, 18:961–968, 2003.